

# Massacre de cerveaux

« L'usage de la calculatrice n'est pas autorisé. »

```
+++++++ [>++++>+++++
+++++++>+++++
+>+++++>+++++
+++++++>+++++
+++++++>+++++
+++++++>+++++
+++++++>+++++
>+<<<<<<<<<-]>>>>>+
++.>>>-.<<<<---.--.<
<+.>-.<++++.<.>-+.<
.>>+++++.<----.<.>+
.--.>.<+.>>+.<.<--
.>>-.--.<----.<+.>+
<.>>+++++.[>]<<.[<]>
>.>>+.>.<<<.>>+.<--
--.>>-.<<<.>.>+.-.
<-.>--.<-----.>.>>
+++.>+.>+++++.>+.
```

*Avertissement : la suite de ce document ne fait pas partie, à proprement parler, de la contribution. Bien qu'elle puisse vous aider à comprendre la page précédente, elle risque aussi de changer radicalement votre opinion sur celle-ci et sur son auteur. Libre à vous de choisir de lire la suite ou non.*

Quel sens donner à cette contribution ? Il y a au moins deux pistes à retenir. La première, celle que vous avez probablement saisie, consiste à y voir une réflexion sur la mise en page d'un document, sur ce que les typographes appellent le gris du texte et à propos duquel Lacroux écrit joliment : « la couleur essentielle de la page, la plus belle, la plus émouvante, c'est le gris du texte, précaire, irréel »<sup>1</sup>. Faut-il voir dans le remplacement des lettres par des symboles une rupture du gris du texte ? Ou au contraire, celui-ci est-il conservé malgré la perte de sens ?

Le sens, j'y viens. Car s'il n'y en avait pas, cette contribution serait ambiguë<sup>2</sup> et ne répondrait pas au sujet fixé. Il nous faut donc s'attarder sur ce prétendu langage de programmation (Turing-complet, pour les initiés) : le *brainfuck*. Il s'agit en fait plutôt d'un jeu ! Imaginez que vous avez une infinité de cases mémoires devant vous (0, 1, 2, etc.), toutes initialisées à 0. Au début, vous vous trouvez en face de la case 0 et vous ne pouvez faire que huit choses, symbolisées par huit caractères.

- '>' vous permet de vous déplacer sur la case suivante.
- '<' vous permet de vous déplacer sur la case précédente.
- '+' vous permet d'ajouter 1 à la case actuelle.
- '-' vous permet de soustraire 1 à la case actuelle.
- '.' affiche à l'écran la valeur de la case actuelle sous la forme du caractère correspondant en ASCII (par exemple 97 correspond à la lettre *a*). Selon l'encodage de votre ordinateur vous pouvez également utiliser l'ASCII étendu, ou l'Unicode.
- ',' lit un caractère ASCII et range sa valeur dans la case actuelle.
- '[' saute au caractère ']' suivant si la case actuelle contient 0.
- ']' saute au caractère '[' précédent si la case actuelle contient une valeur non nulle.

Vous l'avez compris, les deux derniers caractères permettent de coder des boucles, voir même des conditions. Vous avez maintenant toutes les clés en main pour comprendre ma contribution. Mais comme je suis sympathique, je vous donne une version commentée du code !

---

1. *Orthotypographie, Orthographe & Typographie françaises, Dictionnaire raisonné*, Jean-Pierre Lacroux, disponible sous licence libre en ligne à l'adresse [orthotypographie.fr](http://orthotypographie.fr)

2. « ambiguë » pour les partisans de l'orthographe de grand'mère.

```

On charge d'abord quelques nombres en mémoire
+++++ +++ On inscrit 8 en cellule 0
[ Début boucle (tant que la cellule 0 est non nulle)
  >++++ Pour faire 32 en cellule 1
  >+++++ +++++ ++ Pour faire 96 en cellule 2
  >+++++ +++++ +++ Pour faire 104 en cellule 3
  >+++++ +++++ +++++ Pour faire 120 en cellule 4
  >+++++ +++++ Pour faire 72 en cellule 5
  >+++++ +++++ +++++ +++++ +++++ Pour faire 192 en cellule 6
  >+++++ +++++ +++++ +++++ + Pour faire 168 en cellule 7
  >+++++ Pour faire 40 en cellule 8
  >+ Pour faire 8 en cellule 9
  <<<<< <<<<- Décrémente cellule 0
] Fin boucle
Et maintenant on écrit le texte !
>>>>>++++. L (76) depuis cellule 5
>>>>- . apostrophe (39) depuis cellule 8
<<<<<----. u (117) depuis cellule 4
--. s (115) depuis cellule 4
<<<+. a (97) depuis cellule 2
>-. g (103) depuis cellule 3
<++++. e (101) depuis cellule 2
<. espace (32) depuis cellule 1
>-. d (100) depuis cellule 2
+. e (101) depuis cellule 2
<. espace (32) depuis cellule 1
>>+++++. l (108) depuis cellule 3
<----. a (97) depuis cellule 2
<. espace (32) depuis cellule 1
>++. c (99) depuis cellule 2
--. a (97) depuis cellule 2
>. l (108) depuis cellule 3
<++. c (99) depuis cellule 2
>>++. u (117) depuis cellule 4
<. l (108) depuis cellule 3
<--. a (97) depuis cellule 2
>>>- . t (116) depuis cellule 4
--. r (114) depuis cellule 4
<----. i (105) depuis cellule 3
<++. c (99) depuis cellule 2
++. e (101) depuis cellule 2

```

```

<.          espace (32) depuis cellule 1
>>+++++.  n (110) depuis cellule 3
[>]<<.    apostrophe (39) depuis cellule 8
[<]>>.    e (101) depuis cellule 2
>>+.      s (115) depuis cellule 4
+.        t (116) depuis cellule 4
<<<.      espace (32) depuis cellule 1
>>++.     p (112) depuis cellule 3
<----.    a (97) depuis cellule 2
>>-.      s (115) depuis cellule 4
<<<.      espace (32) depuis cellule 1
>.        a (97) depuis cellule 2
>>++.     u (117) depuis cellule 4
-.        t (116) depuis cellule 4
<-.      o (111) depuis cellule 3
>--.     r (114) depuis cellule 4
<----- -. i (105) depuis cellule 3
>+.      s (115) depuis cellule 4
>>++++.  é (195 puis 169 en codage UTF8) depuis cellule 6
>+.      et cellule 7
>+++++ ++. point (47) depuis cellule 8
>++.     nouvelle ligne (10) depuis cellule 9

```